

## Chapter 8

# Differential Equations

This chapter focuses on the use of Maple commands to analyze differential equations. We use Maple's `dsolve` command to find explicit solutions to various first and second order differential equations. Maple is very good at finding explicit solutions for differential equations—when they can be found. However, solutions to many of the important differential equations cannot be found in closed form. In this case, the `numeric` option of `dsolve` can be used to find approximate numerical solutions to differential equations. In addition, the `DEplot` command can be used to plot direction fields, solution curves and phase portraits.

### 8.1 Explicit Solutions

To solve the first order, linear differential equation  $y' + 5y = 2x$ , first enter the equation as

```
> eq1:=diff(y(x),x)+5*y(x)=2*x;
```

$$eq1 := \left(\frac{d}{dx} y(x)\right) + 5y(x) = 2x$$

Notice that you must write `y(x)` and not `y` so that Maple knows `y` is a function of `x`. Then solve the equation using `dsolve`:

```
> dsolve(eq1,y(x));
```

$$y(x) = -\frac{2}{25} + \frac{2x}{5} + e^{(-5x)}\_C1$$

Notice that the symbol `_C1` starts with an underscore, and so multiplies the  $e^{(-5x)}$ . This `_C1` is Maple's notation for the arbitrary constant that occurs in the general solution of a first order differential equation. Also notice that the output is an equation (rather than an expression or a function or an assignment). So you need to use the `rhs` command to assign the right hand side to a label:

```
> ysol:=rhs(%);
```

$$ysol := -\frac{2}{25} + \frac{2x}{5} + e^{(-5x)}\_C1$$

An initial condition such as  $y(-1) = 2$  can be imposed in the following manner.


```
> dsolve({eq1,y(-1)=2},y(x));
```

$$y(x) = -\frac{2}{25} + \frac{2x}{5} + \frac{62}{25} \frac{e^{(-5x)}}{e^5}$$

Note that the equation and the initial condition are enclosed in curly braces. To see a plot of the solution, the following Maple commands can be used:

```
> ysol:=rhs(%):
```

```
> plot(ysol, x=-2..3, -1..4);
```



Notice that we have used the variables  $x$  and  $y$  here. However the independent variable is frequently taken as  $t$  to denote time.

The commands to solve a second (or higher) order differential equation are similar to the above. To find the general solution to the second order differential equation

$$\frac{d^2y}{dt^2} + 6\frac{dy}{dt} - 7y = 1$$

enter the commands

```
> eq2:=diff(y(t),t$2)+6*diff(y(t),t)-7*y(t)=1;
```

$$eq2 := \left(\frac{d^2}{dt^2} y(t)\right) + 6\left(\frac{d}{dt} y(t)\right) - 7y(t) = 1$$

```
> dsolve(eq2,y(t));
```

$$y(t) = e^t \_C2 + e^{(-7t)} \_C1 - \frac{1}{7}$$

Notice that there are two arbitrary constants in the solution,  $\_C1$  and  $\_C2$ . To specify initial conditions such as  $y(1) = 3$  and  $y'(1) = 2$ , enter

```
> dsolve({eq2, y(1)=3, D(y)(1)=2},y(t));
```

$$y(t) = \frac{3e^t}{e} + \frac{1}{7} \frac{e^{(-7t)}}{e^{(-7)}} - \frac{1}{7}$$

Notice that the `diff` command cannot be used to specify an initial condition in `dsolve`. Rather, you must use the `D` operator and function notation.

## 8.2 Direction Fields

When you can't solve a differential equation exactly, what do you do?


A direction field is a useful geometric device that aids in understanding the behavior of solutions to a first order differential equation,  $\frac{dy}{dx} = F(x, y)$ . At each point  $(x, y)$  on a solution  $y = f(x)$ , the slope of the solution curve is  $m = f'(x) = F(x, y)$ . The direction field is a way to display these slopes. At each point  $(x, y)$  draw a small line segment with slope  $m = F(x, y)$ . Then any solution curve will be everywhere tangent to these line segments.

Maple can plot direction fields using the `DEplot` command from the `DEtools` package. To plot the direction field for the differential equation  $y' = x \sin(y)$ , first load the `DEtools` package and define the equation:

```
> with(DEtools):
> deq:=diff(y(x),x)=x*sin(y(x));
      deq :=  $\frac{d}{dx} y(x) = x \sin(y(x))$ 
```


Then use the `DEplot` command:

```
> DEplot(deq, y(x), x=-2..2, y=-1..1);
```



To graph one or more solutions to the differential equation together with the direction field, you need to add the initial conditions. Suppose you want three solution curves with initial conditions  $f(0) = 0.5$ ,  $f(1) = -0.5$  and  $f(-1) = 0.25$ . Then you execute


```
> inits:=[[0,.5], [1,-.5], [-1,.25]];
      inits := [[0, 0.5], [1, -0.5], [-1, 0.25]]
> DEplot(deq, y(x), x=-2..2, y=-1..1, inits);
```



Notice how the solution curves are everywhere tangent to the direction field. In fact, this tangency is the basic idea behind Euler's method, which is a numerical algorithm for calculating approximate solutions to differential equations. Euler's method is discussed in Section 12.28.

Finally, to plot the curves without the direction field, add the option `arrows=None`:

```
> DEplot(deq, y(x), x=-2..2, y=-1..1, inits, arrows=None);
```



NOTE: If the solution curve displays sharp corners or other nonsmooth behavior, you should decrease the stepsize by inserting `stepsize=0.01` or whatever is appropriate as the last option in the `DEplot` command. The default step size is  $1/20$  of the  $x$ -range, which is often too big.

### 8.3 Numerical Solutions

The material in this section is optional. However, since many nonlinear differential equations are virtually impossible to solve (in closed form), students should find the information in this section very useful.

Maple can find numerical approximations to solutions to differential equations. The algorithm that Maple uses is a Fehlberg fourth-fifth order Runge-Kutta method, which is more sophisticated than Euler's method. For example, suppose we want to solve  $y' + \sin(y^2) = 1$ , with  $y(0) = 1$ . If we proceed as in the previous section, we get the following.

```
> deq:=diff(y(t),t)+sin(y(t)^2)=1;
      deq := ( $\frac{d}{dt} y(t)$ ) + sin(y(t)2) = 1
> dsolve({deq,y(0)=1},y(t));
```

$$y(t) = \text{RootOf} \left( t + \int_0^{-Z} \frac{1}{\sin(-a^2) - 1} d_a - \int_0^1 \frac{1}{\sin(-a^2) - 1} d_a \right)$$

Maple is telling us that it cannot evaluate some integral in closed form.

To obtain an approximate solution using `dsolve`, add the `numeric` option. (There must be a numeric initial condition.) Maple's output is then a procedure, which can be used in a manner similar to a Maple function. (For details on procedures, see Chapter 10.)


```
> f:=dsolve({deq,y(0)=1},y(t),numeric);
      f := proc(x_rkf45) ... end proc
```

We have assigned the label `f` to this procedure so that we can use it to find approximate values for the solution at various values of  $t$ . We evaluate it just like a function. Its values at  $t = 0$  and  $t = 0.5$  are:

```
> f(0); f(0.5);
      [t = 0., y(t) = 1.]
      [t = 0.5, y(t) = 1.06203222331694258]
```

To plot the solution we use Maple's `odeplot` command in the `plots` package:

```
> with(plots):
> odeplot(f, [t,y(t)], 0..4, color=black);
```



NOTE: The variables in the list  $[t, y(t)]$  must be the same symbols for the independent and dependent variables that were used when the procedure `f` was defined. Alternatively, the plot can be obtained directly by using Maple's `DEplot` command in the `DEtools` package (possibly with less accuracy unless you decrease the `stepsize`):

```
> with(DEtools):
> DEplot(deq,y(t),t=0..4,[[0,1]], arrows=none, stepsize=.01):
```

## 8.4 Systems of Differential Equations

One of the nicest features of Maple is its ability to analyze systems of differential equations. This section demonstrates the use of several Maple tools for analyzing systems of the form

$$\begin{aligned}x'(t) &= f(x(t), y(t)) \\y'(t) &= g(x(t), y(t))\end{aligned}$$

For example, consider the Lotka-Volterra model given by


```
> de1:=diff(x(t),t)=x(t)*(1-y(t));
      de1 :=  $\frac{d}{dt} x(t) = x(t)(1 - y(t))$ 
> de2:=diff(y(t),t)=0.2*y(t)*(x(t)-1);
      de2 :=  $\frac{d}{dt} y(t) = 0.2y(t)(x(t) - 1)$ 
```

We start by loading the DEtools package.

```
> with(DEtools):
```

The DEplot command can be used to create a direction field plot as follows.


```
> DEplot([de1,de2], [x(t),y(t)], t=0..1, x=0..3, y=0..2,
> scaling=constrained);
```



At each point, the small arrow shows the direction along which  $x$  and  $y$  will change as  $t$  increases.

We can add solution curves to the plot by adding initial conditions. Notice how the two sets of initial data are specified below.

```
> DEplot([de1,de2], [x(t),y(t)], t=0..20, [[x(0)=1.5,y(0)=1.5],
> [x(0)=2,y(0)=2]], stepsize=0.1, scaling=constrained);
```



The solution curves are also called integral curves or phase trajectories, while the direction field plot together with several phase trajectories is called a phase portrait.

We conclude this section by demonstrating how Maple can be used to generate approximate numeric solutions to the system of differential equations. Suppose we want a solution to the Lotka-Volterra system given above corresponding to the initial data  $x(0) = 1.5$  and  $y(0) = 1.5$ . We try to get an exact solution using `dsolve`

```
> sol:= dsolve({de1, de2, x(0)=1.5, y(0)=1.5}, {x(t),y(t)});
      sol :=
```

but Maple can't do it, as indicated by the error message. So we use `dsolve` with the `numeric` option to obtain an approximate solution.

```
> sol:= dsolve({de1, de2, x(0)=1.5, y(0)=1.5}, {x(t),y(t)},
> numeric, output=listprocedure, abserr=0.001):
```

The result is a procedure. We can use this procedure to obtain approximations for our solution at any time value. For example, the approximate values at  $t = 1$  and  $t = 2$  are computed below.

```
> sol(1); sol(2);
[t(1) = 1., x(t)(1) = 0.873670307150112291, y(t)(1) = 1.54975711402056548]
[t(2) = 2., x(t)(2) = 0.524899401190241032, y(t)(2) = 1.45335581162035842]
```

We can also force Maple to extract only the portions that we are interested in. If we want separate functions for  $x(t)$  and  $y(t)$ , then we can proceed as follows:

```
> xsol:=subs(sol,x(t)): ysol:=subs(sol,y(t)):
> [xsol(1), ysol(1)];
[0.873670307150112291, 1.54975711402056548]
```


If we want to see a list of approximate values then we can create a loop.

```
> for i from 0 by 0.2 to 1 do
> [xsol(i), ysol(i)]
> end do;
[1.50000000000000, 1.50000000000000]
```

```
[1.35357000102293944, 1.52578802910667544]
[1.21617049705240476, 1.54322112219716723]
[1.08986906956533436, 1.55261211326470506]
[0.975711526594190914, 1.55452433804239254]
[0.873670307150112291, 1.54975711402056548]
```


It is also possible to plot the approximate solutions obtained from the `numeric` option of `dsolve` by using the `odeplot` command in the `plots` package. We first produce the phase trajectory:

```
> with(plots):
> odeplot(sol, [x(t),y(t)], t=0..20, numpoints=100,
> view=[0..3,0..2], scaling=constrained);
```



(This solution curve can also be produced using the `DEplot` command.) Notice, this plot shows the solution is periodic. To determine the period one can use trial and error on the range of  $t$  in the `odeplot` to see where the plot closes or one can separately plot  $x(t)$  and  $y(t)$ :

```
> odeplot(sol, [[t,x(t)], [t,y(t)]], t=0..30, numpoints=100,
> view=[0..30,0..3]);
```



From this sketch, it appears as though the period of the solution is roughly 15.



## 8.5 Summary

- The `dsolve` command can be used to find exact solutions of single differential equations or systems of differential equations without or with initial conditions. The solution curves can be plotted using `plot` or the `DEplot` command, in the `DEtools` package.
- The `dsolve` command can also be used to find approximate solutions of single or systems of differential equations by adding the `numeric` option. The numeric solution curves can be plotted using the `odeplot` command, in the `plots` package.
- The `DEplot` command, in the `DEtools` package, can be used to plot direction fields, solution curves and phase portraits for single differential equations or systems of differential equations.

## 8.6 Exercises

In Exercises 1-3, use Maple to find the general solution of the differential equations. Then find and plot the particular solution satisfying the initial condition.

1.  $\frac{dy}{dx} = \frac{e^x}{1+y}$  with  $y(0) = 0$ .
2.  $y' + (\sin x)y = \sin(2x)$  with  $y(\pi) = 1$ .
3.  $\frac{dy}{dx} = \frac{xy^2 + x}{x^2 + 1}$  with  $y(2) = -1$ .
4. Plot the direction field of the differential equation  $y' = x^2 - y$  using `DEplot`. On a printout of this plot, hand sketch the solution of this differential equation that passes through the point  $[0, 3]$ . Then plot the solution and the direction field together. (Don't forget the *extra* set of square brackets surrounding initial conditions required by the `DEPlot` command.)
5. Plot the direction fields for each of the differential equations in Exercises 1 - 3 together with the solution curve for the given initial condition.
6. Find the general solution of the differential equation:  $y'' + y' + 5y = 0$ . Then find the particular solution satisfying the initial conditions  $y(0) = 1$  and  $y'(0) = 2$ . Finally plot the particular solution.
7. Solve the equation  $y' + y = 2 \sin(x) \cos(x)$ , with initial condition  $y(0) = 0$ , both exactly and numerically. Use the `display` command to plot both solutions together using the range  $-1 \leq x \leq 10$ . How well does the numerical solution approximate the exact solution?
8. Use Maple to sketch the direction field for the system

$$\begin{aligned}x'(t) &= -0.1x(t) - y(t) \\y'(t) &= x(t) - 0.1y(t)\end{aligned}$$

Sketch the solution trajectories associated with a variety of initial data pairs. What do you notice about the behavior of the solution curves?

9. The purpose of this problem is to describe the phase portraits for the system

$$\begin{aligned}x'(t) &= ax(t) - by(t) \\y'(t) &= bx(t) + ay(t)\end{aligned}$$

for a variety of choices of the parameters  $a$  and  $b$ .

- (a) Start by setting  $b = 1$ . Plot the direction field for  $a = -3, -2, -1, 0, 1, 2, 3$ . What do these phase portraits have in common? How do they differ? Comment on whether a point moves in or out, clockwise or counterclockwise.
- (b) Repeat this process with  $b = -1$ .
- (c) Repeat this process with  $b = 0$ .
- (d) Summarize your results.
10. Show that if  $c_1$  and  $c_2$  are constants then

$$\begin{aligned}x(t) &= c_1 e^{at} \cos(bt) - c_2 e^{at} \sin(bt) \\y(t) &= c_1 e^{at} \sin(bt) + c_2 e^{at} \cos(bt)\end{aligned}$$

solve the system in Exercise 9. Does this reinforce your conclusions from Exercise 9? Explain.

11. Consider the Lotka-Volterra system given by

$$\begin{aligned}x'(t) &= x(t)(1 - y(t)) \\y'(t) &= 0.2y(t)(x(t) - 1)\end{aligned}$$

The phase portrait for this system was sketched earlier in this chapter. From looking at the phase portrait, it appears as though the positive solutions of this system are periodic. Use the `numeric` option of `dsolve` to determine the period of the solutions to this system corresponding to each of the initial data in the following table:

$x(0)$	0.2	0.4	0.6	0.8	1.2	1.4	1.6
$y(0)$	0.2	0.4	0.6	0.8	1.2	1.4	1.6

Why was the initial condition  $(1, 1)$  deliberately omitted from the table?

HINT: You might want to try a `do` loop:

```
> for i from 0.2 by 0.2 to 1.6 do
>   inits:= x(0)=i, y(0)=i;
>   sol:=dsolve({deqs,inits}, {x(t),y(t)}, numeric);
>   plots[odeplot](sol, [x(t),y(t)], t=0..25);
> end do;
```